Code and pseudo code

Ilan Vardi

Appeared in The Mathematica Journal 6 Issue 2 (1996), 66–71.

It is quite common that mathematical results rely on explicit descriptions of algorithms. These are often written in what is called “pseudo code” usually based on a computer language like Pascal or C [2] [8] [9]. The question addressed here is how to check that pseudo code is correct, in other words, how do you get a “pseudo compiler” to check your pseudo code? I propose that a way to do this is to use a high level language such as Mathematica to write a real implementation. The high level and built–in mathematical capabilities of such a language allow you to write pseudo code that is essentially a transcription of the computer code, and so hopefully free of “pseudo bugs.” Acknowledgments. I would like to thank Henry Cejtin and Igor Rivin for helpful comments.

1. A simple example Consider the problem of adding and multiplying positive integers by using their decimal expansions, i.e., describe algorithms that given two strings of digits representing two integers, return the strings of digits corresponding to the sum and product of the numbers.

The input will be sequences of digits [d_n, . . . , d_0]. Concatenation x ⋆ y of sequences will be used along with the unusual convention that [0] ⋆ x returns x. This condition corresponds to the usual convention that omits leading zeroes in decimal expansions.

Addition algorithm: Given as input two sequences of digits [a_m, . . . , a_0] and [b_n, . . . , b_0] return a sequence a ⊕ b: intialize k ⇐ 0 for i = 0, 1, . . . , max(m, n) do

k ⇐ a_i + b_i + ⌊k/10⌋. If i > m, replace a_i with 0 and if i > n, replace b_i with 0 c_i ⇐ k mod 10 return [⌊k/10⌋] ⋆ [c_{max(m,n)}, . . . , c_0]

Multiplication algorithm: Given as input two sequences of digits [a_m, . . . , a_0] and [b_n, . . . , b_0] return a sequence a ⊗ b: initialize c ⇐ [ ] for i = 0, 1, . . . , m and j = 0, 1, . . . , n do c ⇐ c ⊕ ([⌊a_ib_j /10⌋] ⋆ [a_ib_j mod 10, i+j 0, 0, . . . , 0 ]) return c

The Mathematica implementation starts by defining the ⋆ operator

x_~join~y_:= If[x == {0}, y, x~Join~y]

The Mathematica statement

If[condition, statement1, statement2]

means the same as

if condition then statement1 else statement2 The infix notation x~join~y, meaning the same as join[x, y], has been used in order to make the program look more like the mathematical description of the algorithm.

The next thing is to write functions that take a list [d_k, . . . , d_0] and return the last element d_0, and the list without its last element, i.e., [d_k, . . . , d_1].

last[x_]:= {Last[x]}

drop[x_]:= Drop[x, -1]

The addition and multiplication programs are then

a_~plus~b_:=
Block[{c = {}, i, k = 0},
       Do[k = If[i > Length[a], 0, a[[-i]]] + If[i > Length[b], 0, b[[-i]]] +
              Floor[k/10];
          PrependTo[c, Mod[k, 10]],
          {i, 1, Max[Length[a], Length[b]]}];
       Return[{Floor[k/10]}~join~c]
      ]
a_~times~b_:=
Block[{c = {}, i, j, k},
      Do[k = a[[-i]] b[[-j]];
         c = c~plus~({Floor[k/10]}~join~Prepend[Table[0,{i+j-2}], Mod[k,10]]),
         {i, 1, Length[a]}, {j, 1, Length[b]}];
      Return[c]
     ]

Note that Mathematica has a somewhat idiosyncratic syntax, for example, the fact that arrays can be accessed from the right using negative indices, or expressions like

Quotient[#, 10]~join~Mod[#,10]& [last[a] last[b]],

used below. On the other hand, the control structure of the program is identical to the high one in the high level description and once you get used to the peculiarities of the language, the high level description can be directly transcribed from the program.

An even better example is to use a recursive algorithm

Addition algorithm: Given as input two sequences of digits a = [a_m, . . . , a_0] and b = [b_n, . . . , b_0], return a sequence a ⊕ b:

if a = [ ] or b = [ ] then a ⋆ b

else (([a_m, . . . , a_1] ⊕ [⌊(a_0 + b_0)/10⌋]) ⊕ [b_n, . . . , b_1]) ⋆ [(a_0 + b_0) mod 10]

Multiplication algorithm: Given as input two sequences of digits a = [a_m, . . . , a_0] and a = [b_n, . . . , b_0] return a sequence a ⊗ b:

if n = 0 then

if m = 0 then [⌊a_0b_0/10⌋] ⋆ [a_0b_0 mod 10] else b ⊗ a

else ((a ⊗ [b_n, . . . , b_1]) ⋆ [0]) ⊕ (a ⊗ [b_0])

The return command has been omitted from the description as it is assumed that the algorithm returns the last evaluation. This is true of the Mathematica language, so the Return[] statements will be omitted from the programs as well. The Mathematica implementation is

a_~plus~b_:=
If[Length[a] Length[b] == 0,
   a~join~b,
   Block[{sum = Quotient[#, 10]~join~Mod[#,10]& [last[a] + last[b]]},
         ((drop[a]~plus~drop[sum])~plus~drop[b])~join~last[sum]
   ] ]

where Quotient[a, b] means ⌊a/b⌋.

a_~times~b_:=
If[Length[b]==1,
   If[Length[a] == 1,
      Quotient[#, 10]~join~Mod[#,10]& [last[a] last[b]],
      b~times~a],
   ((a~times~drop[b])~join~{0})~plus~(a~times~last[b])
  ]

In this case, one can argue that a better choice of computer language would be Lisp or Scheme since the programs relied on list operations and recursions, the basic paradigms of these languages.

2. Using a lower level language I recently asked an undergraduate class to solve the problem of the previous section, and then to implement their algorithm. It turns out that all of them managed to write correct programs in C or Pascal, but that none of them wrote correct high level descriptions.

The students had a harder time writing the high level description because they wrote low level programs. The following C implementation does in fact follow the first two algorithms of Section 1 closely, but this is hidden by code dealing with input, output, and memory allocation. Of course, this program will run at least 1, 000 times faster than the Mathematica program, but it’s clear that speed is not the point of this exercise. The data structure is an array a[] where a[0] denotes the length of a[]. The main control structure is

#include <stdio.h>

int *plus(int *num_a, int *num_b);
int *times(int *tnum_a, int *tnum_b);

main()
{int x[1001], y[1001], *z, s; /* up to 1000 digits */
 for(x[0] = 1; (x[x[0]]=getchar()-’0’) >= 0; x[0]++); /* first number */
 for(y[0] = 1; (y[y[0]]=getchar()-’0’) >= 0; y[0]++); /* second number */
 z = ((x[x[0]] + ’0’) == ’+’)?plus(x, y):times(x, y); /* add or multiply */
 for(s = 1; s < z[0]; s++) putchar(z[s]+’0’); /* print answer */
}

The addition program first finds the longest input string and allocates memory accordingly

int *plus(int *p_a, int *p_b)
{int *a, *b, *c, size_a, size_b, k;
 size_a = (a = (p_a[0] > p_b[0])?p_b:p_a)[0];
 size_b = (b = (p_a[0] > p_b[0])?p_a:p_b)[0];
 (c = (int *) calloc(size_b+1, sizeof(int)))[0] = size_b + 1;
 k = 0;
 while(size_b > 1) c[size_b] = (k=(--size_a?a[size_a]:0)+b[--size_b]+ k/10)%10;
 c[1] = (k/10)?1:--(c++[0]); /* last digit */
 return(c);
}

The multiplication program uses the fact that the memory allocated by calloc is initialized to zero. Right shifts simply consist of incrementing the array length.

int *times(int *t_a, int *t_b)
{int *c, *temp, j, size_a, size_b, k;
 (c = (int *) calloc(1, sizeof(int)))[0] = 2; /* c = 0 */
 temp = (int *) calloc(2001, sizeof(int));
 for(size_a = t_a[0]-1; size_a; size_a--){
     temp[0] = t_a[0] - size_a + 1;
     for(size_b = t_b[0]-1; size_b; size_b--){
         temp[1] = (k = t_a[size_a] * t_b[size_b])/10;
         temp[2] = k%10;
         temp[0]++; /* left shift */
         c = plus(c, temp);
     }
    }
 if(!c[1]) c[1] = --(c++[0]); /* remove leading zero */
 return(c);
}

3. A non trivial example Writing an algorithm to do addition and multiplication does not require the programming language to have any mathematical features and, as noted above, using Lisp or Scheme would be a good way to go. I will now give a more complicated example where such things as built–in matrix multiplication will be required, so that using a computer algebra system like Macsyma, Maple or Mathematica is advantageous.

Recall that a continued fraction is a generalization of compound fractions like^{14}_{11} = 1^3_{11} 14 11 = 1 3 11 = 1 + 1 11/3

= 1 + 1 3^2_3

= 1 + 1 3 + 1 3/2 = 1 + 1 3 + 1 1^1_2

= 1 + 1 3 + 1 1 + 1 2 .

In general, every rational number p/q can be written in the form p q = a_0 + 1 a_1 + 1 a_2 + . .. + 1 a_r where the a_i, i > 0, are positive integers and for ease of notation this is usually written as p q = [a_0, a_1, . . . , a_r] .

Consider the problem of adding and multiplying the sequence of digits of two continued fraction expansions. Even multiplying a continued fraction by 2 is already a time consuming task based on “Hurwitz rules” [6, Exercise 4.5.3.14], so the general problem was declared to be hopeless by Khinchin, the renowned expert in the field [5], who wrote: “There is, however, another and yet more significant practical demand that the aparatus of continued fractions does not satisfy at all. Knowing the representations of several numbers, we would like to be able, with relative ease, to find the representations of the simpler functions of these numbers (especially, their sum and product). In brief, for an apparatus to be suitable from a practical standpoint, it must admit sufficiently simple rules for arithmetical operations; otherwise, it cannot serve as a tool for calculation. We know how convenient systematic fractions are in this respect. On the other hand, for continued fractions there are no practically applicable rules for arithmetical operations; even the problem of finding the continued fraction for a sum from the continued fraction representing the addends is exceedingly complicated, and unworkable in computational practice.”

The problem was solved in a little known paper of Marshall Hall [4], but an actual description of the addition and multiplication algorithm was not given until the work of R.W. Gosper [1] [3] [6, Exercise 4.5.3.15] [7, p. 78] based on three ideas. The first idea is that a computation like 3x, where x is given as a continued fraction, will rapidly lead to more complicated forms. For example, computing 3 ·^{14}_{11} 3 [1, 3, 1, 2] = 3 ( 1 + 1 [3, 1, 2] ) = 3 [3, 1, 2] + 1 [3, 1, 2] and so forth. Gosper’s idea was not to try to simplify these forms, but to analyze the worst that can happen. It’s not hard to see that you always get a linear fractional form (∗) ax + b cx + d , where a, b, c, d are integers, and x is given as a continued fraction.

Gosper’s second idea was to consider the x as a formal symbol which could input continued fraction digits into (∗). In other words, think of x not as representing a rational number, but as a symbolic quantity that transforms as (∗∗) x → q + 1 x , where q represents the next continued fraction digit. It remains to see how (∗) transforms under (∗∗). A simple computation shows that ax + b cx + d → aqx + bx + a cqx + dx + c .

So if the form (∗) is represented by a matrix (_{a b} c d ) , then the transformation law corresponding to inputting a digit is ( a b c d ) → ( aq + b a cq + d c ) = ( a b c d ) ( q 1 1 0 ) . The third idea is to realize that you can output continued fraction coefficients without having complete knowledge of x. Gosper’s original example was

70x + 29 12x + 5

If x > 0, then 29 5 ≤ 70x + 29 12x + 5 < 70 12 . This means that 70x + 29 12x + 5

= 5 +

10x + 4 12x + 5 . Similarly, 12 10 ≤ 12x + 5 10x + 4 < 5 4 so 12x + 5 10x + 4

= 1 +

2x + 1 10x + 4 , and 10x + 4 2x + 1

= 4 + 2x 2x + 1 . This means that for any x > 0 you get a continued fraction expansion

70x + 29 12x + 5

= 5 + 1 1 + 1 4 + 1 (2x + 1)/x

In other words, you are able to output 3 continued fraction coefficients of the form (10x+25)/(2x+1) without knowing too much about x. (Note that you cannot output a fourth coefficient since (2x + 1)/x → 2 + 1/x does not give the correct coefficient unless x > 1.)

In the case when x represents a sequence of continued fraction coefficients, the situation is even better because the coefficients are greater or equal to one. Now, since one is assuming that there is a continuing sequence of continued fraction coefficients, the further assumption that x > 1 can be made, so being able to output a continued fraction corresponds to checking that for some integer n n ≤ a c ≤ a + b c + d < n + 1 , in other words, that ⌊ a c ⌋ = ⌊ a + b c + d ⌋ , and when this happens the output will be the common value q = ⌊a/c⌋. The point is that is always happens, given a sufficient number of digits of x.

It remains to see what happens to the form after this has been output. This is 1 ax + b cx + d − q = cx + d ax − cqx + b − dq ,

and the corresponding matrix transformation is ( a b c d ) → ( c d a − cq b − dq ) = ( 0 1 1 −q ) ( a b c d ) .

All these steps can be combined into an algorithm but first an algorithm to compute the continued fraction of a rational number must be written. This essentially follows the^{14}_{11} example (ordinary concatenation [0] ⋆ [x_0, . . . , x_r] = [0, x_0, . . . , x_r] will be used from now on).

Algorithm to compute the continued fraction expansion of a rational number: Given a rational number q, return a sequence f(q): if q = 1/0 then [ ] else [⌊q⌋] ⋆ f(1/(q − ⌊q⌋))

Algorithm to compute the continued fraction expansion of (ax+b)/(cx+d): Given a matrix (_{a b} c d ) and the continued fraction expansion x = [x_0, . . . , x_r], return the sequence f( (_{a b} c d ) , x): if r = 0 then f((a x_0 + b)/(c x_0 + d)).

else if a, b, c, d > 0 and ⌊_a c ⌋ = ⌊_{a+b} c+d ⌋ then [⌊a/c⌋] ⋆ f ((_{0 1} 1 −⌊a/c⌋ )(_{a b} c d ) , x ) (Output a digit.)

else f ((_{a b} c d ) (_{x0 1} 1 0 ) , [x_1, . . . , x_r] ) (Input a digit.)

In Mathematica these algorithms are

ContinuedFraction[q_]:=
If[q == 1/0,{},{Floor[q]}~Join~ContinuedFraction[1/(q-Floor[q])]

ContinuedFraction[matrix_, x_]:=
If[Length[x] == 1,
   ContinuedFraction[Divide @@ (matrix . {First[x], 1})],
   Block[{q = Positive[Min[matrix]] && Quotient @@ (matrix . {{1, 1}, {0, 1}})},
         If[q && Min[q] == Max[q],
            {#}~Join~ContinuedFraction[{{0, 1}, {1, -#}} . matrix, x]& [First[q]],
            ContinuedFraction[matrix . {{First[x], 1}, {1, 0}}, Rest[x]]
   ] ] ]

Since division by zero has been allowed the line

Off[Power::infy]

is useful in order to avoid annoying warning messages.

Note the similarity in the control structure of the program and the pseudo code. It should be clear that it was essentially a transcription of the computer program.

Finally, the algorithms for addition and multiplication can be described. Just as before computing x + y or xy leads to more complicated expressions. The worst expression you get is a “bilinear fractional form” (∗ ∗ ∗) axy + bx + cy + d exy + fx + gy + h . Treating x and y as formal variables, this can be represented by an ordered pair of matrices (or tensor) (( a b c d ) , ( e f g h )) . Letting y → q + 1/y in (∗ ∗ ∗) gives a transformation for the tensor (( a b c d ) , ( e f g h )) → (( a b c d ) , ( e f g h )) ( q 1 1 0 ) , where ordinary matrix multiplication is used on each component. Similarly, letting x → q + 1/x in (∗ ∗ ∗) corresponds to (( a b c d ) , ( e f g h )) → ( q 1 1 0 ) (( a b c d ) , ( e f g h )) .

Just as before, one uses an Euclidean algorithm to output coefficients, but this time there won’t

be a guarantee that intermediate values will be greater than one. A tricky part of the algorithm is

to decide whether to choose x or y to input the next digit. A direct solution to this is to alternate

between them and this requires a componentwise transpose (( a b c d ) , ( e f g h ))_T = (( a c b d ) , ( e g f h )) , corresponding to switching x and y in (∗ ∗ ∗).

Algorithm to compute the continued fraction of (axy+bx+cy+d)/(exy+fx+gy+h):

Given as input a tensor ((_{a b} c d ) , (_{e f} g h )) and continued fraction expansions x = [x_0, . . . , x_r],

y = [y_0, . . . , y_s], return a sequence f (((_{a b} c d ) , (_{e f} g h )) , x, y ) : if s = 0 then f (((_{a b} c d ) , (_{e f} g h )) (_{y0} 1 ) , x ) . else if a, b, c, d, e, f, g, h > 0 and ⌊a/e⌋ = ⌊b/f⌋ = ⌊c/g⌋ = ⌊d/h⌋ then [⌊a/e⌋] ⋆ f ((_{0 1} 1 −⌊a/e⌋ ) ((_{a b} c d ) , (_{e f} g h )) , x, y ) (Output a digit.)

else f ({( (_{a b} c d ) , (_{e f} g h )) (_{y0 1} 1 0 )}T , [y_1, . . . , y_s], x ) (Input a digit and switch x, y.)

This algorithm allows one to add and multiply continued fractions according to the rules x ⊕ y = f (((_{0 1} 1 0 ) , (_{0 0} 0 1 )) , x, y ) , x ⊗ y = f (((_{1 0} 0 0 ) , (_{0 0} 0 1 )) , x, y ) . The program is simplified by the fact that ordinary Mathematica matrix multiplication for gen-

eralized matrices behaves in the same way as the componentwise matrix multiplication defined

above.

ContinuedFraction[tensor_, x_, y_]:=
If[Length[y] == 1,
   ContinuedFraction[tensor.{First[y], 1}, x],
   Block[{q = Positive[Min[tensor]] && Quotient @@ tensor},
          If[q && Min[q] == Max[q],
             {#}~Join~ContinuedFraction[{{0, 1},{1, -#}}.tensor, x, y]& [q[[1,1]]],
              ContinuedFraction[Transpose /@ (tensor.{{First[y], 1},{1,0}}),
                                Rest[y], x]]
   ] ]

ContinuedFractionPlus[x_, y_]:=
ContinuedFraction[{{{0,1},{1,0}},{{0,0},{0,1}}}, x, y]

ContinuedFractionTimes[x_, y_]:=
ContinuedFraction[{{{1,0},{0,0}},{{0,0},{0,1}}}, x, y]

Remark 1: Subtraction and division can be also be computed in this way by using the formulas x ⊖ y = f (((_{0 1} −1 0 ) , (_{0 0} 0 1 )) , x, y ) , x ⊘ y = f (((_{0 1} 0 0 ) , (_{0 0} 1 0 )) , x, y ) . Remark 2: The real advantage of Gosper’s method is that it allows you to add and multiply numbers with known continued fraction expansions, for example, e = [2, 1, 2, 1, 1, 4, 1, 1, 6, . . .] and ϕ = √ 5+1 2 = [1, 1, 1, . . .]. The above algorithms are easily modified to compute quantities like e + ϕ where the digit input would be the functions

e(n) = 2 if n = 0 2(n + 1)/3 if n ≡ 2 (mod 3) 1 otherwise, and ϕ(n) = 1 .

Remark 3: There are problems with continued fraction arithmetic. For example, the computation [1, 2, 2, 2, . . .] ⊗ [1, 2, 2, 2, . . .] corresponding to √ 2 × √ 2, will not return a single correct continued fraction coefficient or return [2, a], where a is a large integer, depending on whether one uses an odd or even length input. This is slightly different from the decimal case.

References

[1] M. Beeler, R.W. Gosper, and R. Schroeppel, HAKMEM, A.I. Lab Memo # 239, M.I.T.

1972.

[2] G.H. Gonnet and R. Baeza–Yates, Handbook of Algorithms and Data Structures, Second Edition, Addison–Wesley 1991.

[3] R.W. Gosper, Continued Fraction Arithmetic, preprint 1976.

[4] M. Hall, Jr., On the sum and product of continued fractions, Annals of Math. 48 (1947), 966–993.

[5] A. Ya. Khinchin, Continued Fractions, University of Chicago Press, Chicago 1964. [6] D.E. Knuth, The Art of Computer Programming, Vol. 2, Seminumerical Algorithms, Addison–Wesley 1981.

[7] S. Levy, Hackers: heroes of the computer revolution, Doubleday, 1984. [8] R. Sedgewick, Algorithms, Second Edition, Addison–Wesley 1988. [9] R. Sedgewick, Algorithms in C, Addison–Wesley 1990.